Lezione 3 (parte seconda)

Enrico Bertolazzi

```
> restart;
> with(plots):
```
Warning, the name changecoords has been redefined

```
> # punti di passaggio della spline
  x0, x1, x2, x3 := 0, 1, 3, 4 ;
  y0, y1, y2, y3 := 1, -1, 3, 1 ;
```
$$x0, x1, x2, x3 := 0, 1, 3, 4 \tag{1}$$
$$y0, y1, y2, y3 := 1, -1, 3, 1$$

```
> # definizione dei segmenti di cubica e loro derivate
  S   := (k,x) -> a[k]+b[k]*x+c[k]*x^2+d[k]*x^3 ;
  DS  := unapply(diff(S(k,x),x),(k,x)) ;
  DDS := unapply(diff(DS(k,x),x),(k,x)) ;
```
$$S := (k, x) \rightarrow a_k + b_k x + c_k x^2 + d_k x^3 \tag{2}$$
$$DS := (k, x) \rightarrow b_k + 2 c_k x + 3 d_k x^2$$
$$DDS := (k, x) \rightarrow 2 c_k + 6 d_k x$$

```
> # condizioni di passaggio per i punti di interpolazione
  eqint := { S(1,x0)=y0, S(1,x1)=y1,
             S(2,x1)=y1, S(2,x2)=y2,
             S(3,x2)=y2, S(3,x3)=y3 } ;
```
$$eqint := \{a_3 + 3 b_3 + 9 c_3 + 27 d_3 = 3, a_3 + 4 b_3 + 16 c_3 + 64 d_3 = 1, a_1 = 1, a_1 + b_1 + c_1 + d_1 = -1, a_2 + b_2 + c_2 + d_2 = -1, a_2 + 3 b_2 + 9 c_2 + 27 d_2 = 3\} \tag{3}$$

```
> # condizioni di continuita per la derivata prima
  eqd1 := { DS(1,x1) = DS(2,x1), DS(2,x2) = DS(3,x2) } ;
```
$$eqd1 := \{b_1 + 2 c_1 + 3 d_1 = b_2 + 2 c_2 + 3 d_2, b_2 + 6 c_2 + 27 d_2 = b_3 + 6 c_3 + 27 d_3\} \tag{4}$$

```
> # condizioni di continuita per la derivata seconda
  eqd2 := { DDS(1,x1) = DDS(2,x1), DDS(2,x2) = DDS(3,x2) } ;
```
$$eqd2 := \{2 c_1 + 6 d_1 = 2 c_2 + 6 d_2, 2 c_2 + 18 d_2 = 2 c_3 + 18 d_3\} \tag{5}$$

```
> # condizioni al contorno per spline rilassate
  bc := { DS(1,x0) = 0, DDS(3,x3) = 0 } ;
```
$$bc := \{b_1 = 0, 2 c_3 + 24 d_3 = 0\} \tag{6}$$

```
> # insieme delle incognite del problema
  incognite := {seq(op([a[k],b[k],c[k],d[k]]),k=0..3)};
```
(7)

$$incognite := \{c_3, d_3, a_1, b_1, c_1, d_1, a_2, b_2, c_2, d_2, a_3, b_3, a_0, b_0, c_0, d_0\} \quad \textbf{(7)}$$

```
> # insieme delle equazioni
  equazioni := {op(eqint),op(eqd1),op(eqd2),op(bc)} ;
```

$$equazioni := \{b_1 = 0, a_3 + 3\, b_3 + 9\, c_3 + 27\, d_3 = 3, a_3 + 4\, b_3 + 16\, c_3 + 64\, d_3 = 1, b_1 + 2\, c_1 + 3\, d_1 = b_2 + 2\, c_2 + 3\, d_2, b_2 + 6\, c_2 + 27\, d_2 = b_3 + 6\, c_3 + 27\, d_3, a_1 = 1, a_1 + b_1 + c_1 + d_1 = -1, a_2 + b_2 + c_2 + d_2 = -1, a_2 + 3\, b_2 + 9\, c_2 + 27\, d_2 = 3, 2\, c_1 + 6\, d_1 = 2\, c_2 + 6\, d_2, 2\, c_2 + 18\, d_2 = 2\, c_3 + 18\, d_3, 2\, c_3 + 24\, d_3 = 0\} \quad \textbf{(8)}$$

```
> # risolve il sistema di equazioni
  res := solve(equazioni,incognite);
```

$$res := \left\{b_1 = 0, d_3 = \frac{32}{29}, c_1 = -\frac{144}{29}, c_3 = -\frac{384}{29}, c_2 = \frac{219}{29}, d_1 = \frac{86}{29}, b_3 = \frac{1446}{29}, a_3 = -\frac{1659}{29}, b_2 = -\frac{363}{29}, a_2 = \frac{150}{29}, d_2 = -\frac{35}{29}, a_0 = a_0, b_0 = b_0, c_0 = c_0, d_0 = d_0, a_1 = 1\right\} \quad \textbf{(9)}$$

```
> # costruzione dei tratti di cubica
  S1 := subs(res,S(1,x)) ;
  S2 := subs(res,S(2,x)) ;
  S3 := subs(res,S(3,x)) ;
```

$$S1 := 1 - \frac{144}{29}x^2 + \frac{86}{29}x^3 \quad \textbf{(10)}$$

$$S2 := \frac{150}{29} - \frac{363}{29}x + \frac{219}{29}x^2 - \frac{35}{29}x^3$$

$$S3 := -\frac{1659}{29} + \frac{1446}{29}x - \frac{384}{29}x^2 + \frac{32}{29}x^3$$

```
> # composizione dei tratti in una unica funzione
  ST := piecewise(x >= x0 and x < x1, S1,
                  x >= x1 and x < x2, S2,
                  x >= x2, S3) ;
```

$$ST := \begin{cases} 1 - \frac{144}{29}x^2 + \frac{86}{29}x^3 & -x \le 0 \text{ and } x < 1 \\ \frac{150}{29} - \frac{363}{29}x + \frac{219}{29}x^2 - \frac{35}{29}x^3 & 1 - x \le 0 \text{ and } x < 3 \\ -\frac{1659}{29} + \frac{1446}{29}x - \frac{384}{29}x^2 + \frac{32}{29}x^3 & 3 \le x \end{cases} \quad \textbf{(11)}$$

```
> A := plot(ST,x=x0..x3):
  B := plot([[x0,y0],[x1,y1],[x2,y2],[x3,y3]],
            style=POINT,symbolsize=30):
  display([A,B]);
```




```
> # costruzione della spline con le direttive maple
  spline([x0,x1,x2,x3],[y0,y1,y2,y3],x,cubic);
```

$$\begin{cases} 1 - 3x + x^3 & x < 1 \\ 3 - 9x + 6x^2 - x^3 & x < 3 \\ -51 + 45x - 12x^2 + x^3 & \text{otherwise} \end{cases} \quad \textbf{(12)}$$

```
> # confronto della approssimazione spline
  # e polinomiale nel problema di Runge
  f    := x -> 1/(1+x^2) ;
  N    := 20 ;
  a, b := -5, 5 ;
  X    := [seq(a+i*(b-a)/N,i=0..N)] :
  Y    := [seq(f(X[i+1]),i=0..N)]:
  p    := interp(X,Y,x) :
  pp   := spline(X,Y,x,cubic) :
  A    := plot([f(x),p,pp],x=a..b,y=-0.6..1.2) :
  B    := plot([seq([X[i],Y[i]],i=1..N+1)],style=POINT,symbolsize=20)
  :
  display([A,B]) ;
```

$$f := x \rightarrow \frac{1}{1+x^2}$$

$$N := 20$$

$$a, b := -5, 5$$


1.2
1
0.8
0.6
0.4
0.2
−0.2
−0.4
−0.6
y
x
−4
−2
2
4